

Wstęp

W zastosowaniach spotka się równania lub układy równań, których rozwiązaniem jest liczba (lub skończony zbiór liczb). Przykładem równania nieliniowego z jedną niewiadomą jest równanie kwadratowe, $ax^2 + bx + c = 0$, gdzie $a \neq 0$. W tym przypadku znane są wzory na rozwiązanie, czyli wyrażenia, które w *skończonej liczbie kroków* przy pomocy działań arytmetycznych (+, −, *, /) oraz pierwiastkowania dowolnego stopnia ($\sqrt[n]{}$) wykonanych na współczynnikach równania $\{a, b, c\}$ dają to rozwiązanie. Dla równania kwadratowego są to dobrze znane formuły:

$$x_1 = \frac{-b + \sqrt{b^2 - 4ac}}{2a}, \quad x_2 = \frac{-b - \sqrt{b^2 - 4ac}}{2a}. \quad (1)$$

Ale już dla równania trzeciego stopnia

$$ax^3 + bx^2 + cx + d = 0,$$


wzory są na tyle skomplikowane (*wzory Cardano*), że na ogół nie ma sensu ich stosować do obliczeń chemicznych. Aby to zilustrować rozważmy równanie trzeciego stopnia $x^3 + x + 1 = 0$. Ze wzorów Cardano otrzymujemy jeden z pierwiastków w następującej postaci:

$$x_1 = -\sqrt[3]{\frac{2}{3(-9 + \sqrt{93})}} + \sqrt[3]{\frac{-9 + \sqrt{93}}{18}}.$$

Przybliżona wartość numeryczna to $x_1 \approx -0.682328$. Równanie to posiada jeszcze dwa pierwiastki zespolone, których przybliżone wartości to $x_{2,3} \approx 0.341164 \pm 1.6154i$. Równanie czwartego stopnia może mieć do czterech pierwiastków, a piątego stopnia – do pięciu (różnych) pierwiastków.^[1]

Wiadomo obecnie, że nie można podać ogólnego wzoru na rozwiązanie równania wielomianowego stopnia większego od 5. Ponadto, oprócz równań algebraicznych (wielomianowych) mamy równania niealgebraiczne, np. równanie $x - \cos x = 0$, które także nie posiadają „wzorów” (w powyższym rozumieniu) na rozwiązania. Istnieją jednak metody numeryczne, które pozwalają lokalizować zera wielomianów czy dowolnych równań nieliniowych z wymaganą dokładnością.

Problemy określenia składu ilościowego reagentów w stanie równowagi chemicznej prowadzą właśnie do konieczności rozwiązywania równań lub układów równań nieliniowych. Na przykład, gdy słaby kwas jednoprotonowy HA dysocjuje w wodzie,



a jego formalne stężenie wynosi c_0 , to w podstawowym kursie chemii wyprowadza się równanie na stężenie kationów wodorowych $x = [\text{H}^+]$

¹ Z teorii wielomianów wynika, że równanie algebraiczne stopnia $n \geq 1$

$$a_n x^n + a_{n-1} x^{n-1} + \dots + a_1 x + a_0 = 0, \quad (\text{gdzie } a_n \neq 0),$$

może mieć co najwyżej n pierwiastków (różnych), które w ogólności są zespolone. Może też nie mieć żadnych pierwiastków w dziedzinie liczb rzeczywistych $\mathbb{R}$, na przykład $x^2 + 1 = 0$ nie posiada żadnego rozwiązania w zbiorze $\mathbb{R}$, ale posiada je w zbiorze $\mathbb{C}$, gdyż $x^2 + 1 = (x - i)(x + i)$. W tym przypadku równanie posiada tylko pierwiastki zespolone: $i, -i \in \mathbb{C}$.

$$x^2 + K_a x - c_0 K_a = 0,$$

gdzie K_a to stała równowagi powyższej reakcji (zwana też *stałą kwasowości* bądź *stałą dysocjacji kwasowej*). Ale dla bardzo słabych kwasów ($K_a \leq 10^{-5}$) i bardzo rozcieńczonych roztworów ($c_0 \leq 10^{-6}$ M) szkolny opis jest niepoprawny, bo nie uwzględnia w bilansie jonów H^+ , które powstają także w wyniku autojonizacji wody. Należy zatem jeszcze wziąć pod uwagę reakcję $H_2O \rightleftharpoons H^+ + OH^-$ co prowadzi do następującego równania na stężenie kationów H^+ :

$$x^3 + K_a x^2 - (K_w + c_0 K_a)x - K_a K_w = 0. \quad (2)$$

Ogólnie problem można sformułować następująco:

Dla danej funkcji $f: [a, b] \rightarrow \mathbb{R}$ szukamy takiej liczby $x^ \in \mathbb{R}$, że $f(x^*) = 0$.*

Przykładowo, jeżeli $f(x) = x^2 - 5$, to wtedy równanie $x^2 - 5 = 0$ prowadzi do poszukiwania pierwiastka kwadratowego $x^* = \sqrt{5}$. Jeżeli $f(x) = x - \cos x$, wtedy mamy równanie $x - \cos x = 0$ (lub $x = \cos x$). Dla $f(x) = x^5 - 3x^4 + 5x - 2$ równanie $f(x) = 0$ jest równaniem algebraicznym (wielomianowym) piątego stopnia

$$x^5 - 3x^4 + 5x - 2 = 0.$$

Jak już wspomniano równania takie pojawiają się w wielu zastosowaniach, np. w problemach równowag kwasowo-zasadowych.

Numeryczne metody znajdowania pierwiastków równań nieliniowych to głównie metody iteracyjne. Oznacza to, że tworzymy wg pewnego algorytmu ciąg liczb rzeczywistych $\{x_n\}_{n=0}^{\infty}$, o którym można udowodnić, że jest zbieżny do szukanego rozwiązania x^*

$$\lim_{n \rightarrow \infty} x_n = x^*.$$

Oczywiście w praktyce obliczamy tylko skończoną liczbę wyrazów takiego ciągu co daje numeryczne przybliżenie rozwiązania interesującego nas równania.

Metoda bisekcji (połowienia)

Metoda bisekcji opiera się na twierdzeniu matematycznym zwanym **własnością Darboux**:

Jeśli funkcja ciągła na przedziale $[a, b]$ przyjmuje na końcach tego przedziału wartości o przeciwnych znakach, to funkcja ta posiada miejsce zerowe w przedziale (a, b) .

Można to bardziej formalnie zapisać tak:

Jeżeli $f \in C([a, b])$ oraz $f(a)f(b) < 0$, to istnieje liczba $x^ \in (a, b)$ taka, że $f(x^*) = 0$.*

Opis metody bisekcji:

- 1) oblicz środek przedziału: $c = (a + b) / 2$,
- 2) jeżeli $f(c) = 0$, to $x^* = c$ jest szukanym pierwiastkiem i procedurę kończymy,
- 3) jeżeli $f(c) \neq 0$, to wybieramy ten podprzedział spośród $[a, c]$ i $[c, b]$, w którym funkcja zmienia znak na końcach przedziału,
- 4) jeżeli nie uzyskaliśmy żądanej dokładności, to wracamy do punktu 1).

Metody Matematyczne w Procesach Przemysłowych, Notatki do wykładu,

Wykład 3, 4 (równowagi chemiczne, rozwiązywanie równań nieliniowych)

Przykład 1. Szukamy rozwiązania równania $x - \cos x = 0$. Wykonanie procedury bisekcji z początkowym przedziałem $[a, b] = [0, 2]$ daje po kilkunastu iteracjach (powtórzeniach) wartość $x^* = 0,739085$ jako przybliżoną wartość rozwiązania równania $x = \cos x$. Poniżej jest pokazana realizacja w VBA oraz przy pomocy formuł arkusza kalkulacyjnego.

tmp.xlsm - Arkusz1 (Code)							
(General)							
<pre> Function f(x As Double) As Double f = x - Cos(x) End Function Sub bisekcja() Dim a As Double Dim b As Double Dim c As Double Dim x As Double Dim eps As Double a = 0 b = 2 eps = 0.00001 Do While Abs(b - a) > eps c = (a + b) / 2 If f(c) = 0 Then Exit Do If f(a) * f(c) < 0 Then b = c Else a = c End If Loop MsgBox c End Sub </pre>							
	A	B	C	D	E	F	G
1							
2							
3							
4	n	a	c=(a+b)/2	b	f(a)	f(c)	f(b)
5	0	0	1	2	-1	0,459698	2,416147
6	1	0	0,5	1	-1	-0,37758	0,459698
7	2	0,5	0,75	1	-0,37758	0,018311	0,459698
8	3	0,5	0,625	0,75	-0,37758	-0,18596	0,018311
9	4	0,625	0,6875	0,75	-0,18596	-0,08533	0,018311
10	5	0,6875	0,71875	0,75	-0,08533	-0,03388	0,018311
11	6	0,71875	0,734375	0,75	-0,03388	-0,00787	0,018311
12	7	0,734375	0,742188	0,75	-0,00787	0,005196	0,018311
13	8	0,734375	0,738281	0,742188	-0,00787	-0,00135	0,005196
14	9	0,738281	0,740234	0,742188	-0,00135	0,001924	0,005196
15	10	0,738281	0,739258	0,740234	-0,00135	0,000289	0,001924
16	11	0,738281	0,73877	0,739258	-0,00135	-0,00053	0,000289
17	12	0,73877	0,739014	0,739258	-0,00053	-0,00012	0,000289
18	13	0,739014	0,739136	0,739258	-0,00012	8,47E-05	0,000289
19	14	0,739014	0,739075	0,739136	-0,00012	-1,7E-05	8,47E-05
20	15	0,739075	0,739105	0,739136	-1,7E-05	3,36E-05	8,47E-05

W lewym panelu znajduje się kod VBA, które główną pętlą jest

```
Do While Abs(b-a) > eps
```

```
...
```

```
...
```

```
Loop
```

Warunkiem kontynuacji iteracji tej pętli jest „Abs(b-a) > eps”, co oznacza, że iteracje są wykonywane dopóki długość przedziału [a, b] jest większa od eps (w tym przykładzie eps = 0,0001). W momencie gdy granice przedziału zbliżą się na odległość mniejszą bądź równą eps, pętla zostaje przerwana i wypisujemy instrukcją „MsgBox c” wartość przybliżonego rozwiązania. (Można też wypisywać środek przedziału: MsgBox (a+b) / 2). Pętla może być też przerwana w sytuacji, gdy tak się zdarzy, że akurat trafi się dokładne (numerycznie) rozwiązanie:

```
If f(c) = 0 Then Exit Do
```

W prawym panelu pokazana jest realizacja metody bisekcji w oparciu o formuły arkusza kalkulacyjnego. Formuły w komórkach są następujące:


Komórka (adres)	Formuła
C5	=(B5+D5)/2
E5	=B5-COS(B5)
F5	=C5-COS(C5)
G5	=D5-COS(D5)
B6	=JEŻELI(E5*F5 < 0;B5;C5)
D6	=JEŻELI(E5*F5 < 0;C5;D5)

Po wprowadzeniu powyższych formuł należy je skopiować („przeciąganiem” w dół), na przykład do wiersza nr 20. W tym przypadku będzie to oznaczało wykonanie 15 kroków metody bisekcji (pierwszy obliczenia znajdują się bowiem w wierszu nr 6).

Metoda Newtona

Obliczanie rozwiązań równań nieliniowych (zarówno układów jak i pojedynczych) *metodą Newtona* jest chyba najważniejszą i najczęściej używaną (zwłaszcza dla układów równań).^[2] W przypadku jednowymiarowym zwana jest także metodą stycznych ze względu na jej elegancką interpretację geometryczną. Na ogół iteracje są zbieżne dopiero, gdy punkt startu będzie wybrany dostatecznie blisko szukanego (a zatem nieznanego) rozwiązania.

Idea jest następująca. Wybieramy punkt startu x_0 możliwie najbliżej poszukiwanego rozwiązania, a następnie prowadzimy styczną do wykresu w tym punkcie $(x_0, f(x_0))$. Styczna ta przecina oś Ox w jakimś punkcie, który przyjmujemy za kolejne przybliżenie szukanego pierwiastka. Następnie możemy powtórzyć ten krok. Idea ta jest zilustrowana na rysunku poniżej.



Rys. 1. Ilustracja metody Newtona. Rysujemy styczną do wykresu dla punktu x_0 i jako kolejne przybliżenie miejsca zerowego (x^*) przyjmujemy punkt przecięcia tej stycznej z osią Ox – na rysunku jest to punkt x_1 . Proces ten powtarzamy, aż uzyskamy zadowalającą dokładność rozwiązania.

Prosta która przechodzi przez punkt $(x_0, f(x_0))$ ma równanie $y = a(x - x_0) + f(x_0)$. Ale ma to być styczna, więc jej współczynnik kierunkowy a jest równy pochodnej funkcji $y = f(x)$ w punkcie $x = x_0$, czyli $a = f'(x_0)$ co daje $y = f'(x_0) \cdot (x - x_0) + f(x_0)$. Przyrównując do zera równanie na styczną, $f'(x_0) \cdot (x - x_0) + f(x_0) = 0$, i rozwiązując to równanie względem x otrzymamy podstawowy krok metody stycznych

$$x_1 = x_0 - \frac{f(x_0)}{f'(x_0)}.$$

² Gdy metoda ta jest stosowana do układów, to w literaturze często jest nazywana *metodą Newtona–Raphsona*.

Metody Matematyczne w Procesach Przemysłowych, Notatki do wykładu,

Wykład 3, 4 (równowagi chemiczne, rozwiązywanie równań nieliniowych)

Powtarzając ten schemat, ale teraz dla x_1 otrzymujemy $x_2 = x_1 - f(x_1) / f'(x_1)$. Możemy tę procedurę kontynuować co prowadzi do następującego ciągu przybliżeń:

$$x_{n+1} = x_n - \frac{f(x_n)}{f'(x_n)} \quad \text{dla } n=0, 1, 2, \dots \quad (3)$$

Z punktu widzenia implementacji metody Newtona podstawowy problem polega na tym, że należy dostarczyć do procedury także pochodną równania, $f'(x)$.

Poniżej zaprezentowano prosty program w języku VBA realizujący metodę Newtona dla równań nieliniowych z jedną niewiadomą. W szczególności program został użyty do obliczenia pierwiastka równania $x^3 + 2x^2 + 5x + 1 = 0$. Widać, że do uruchomienia procedury musimy określić lewą stronę, czyli $f(x) = x^3 + 2x^2 + 5x + 1$ oraz jej pochodną $Df(x) = f'(x) = 3x^2 + 4x + 5$. Jako punkt startu wybrano $x_0 = 5,0$. Po dziesięciu iteracjach otrzymujemy przybliżenie rozwiązania $x^* \approx -0,216757$. Jednak analiza wyników działania programu pokazuje, że wartość tą uzyskano już w iteracji numer 7.

	A	B	C
1			
2		Metoda Newtona	
3	Przykład:	$x^3+2x^2+5x+1=0$	
4			
5		n	x_n
6		0	5
7		1	2,99
8		2	1,606704
9		3	0,649846
10		4	0,044377
11		5	-0,19213
12		6	-0,21656
13		7	-0,21676
14		8	-0,21676
15		9	-0,21676
16		10	-0,21676
17		11	-0,21676
18		12	-0,21676
19		13	-0,21676
20		14	-0,21676
21		15	-0,21676
22			

```

Function f(x As Double) As Double
f = x ^ 3 + 2 * x ^ 2 + 5 * x + 1
End Function

Function Df(x As Double) As Double
Df = 3 * x ^ 2 + 4 * x + 5
End Function

Sub metoda_newtona_1D()
Dim x As Double
Dim n As Integer
Dim k As Integer

k = 6
x = Cells(k, 3)


For n = 1 To 15
    x = x - f(x) / Df(x)
    k = k + 1
    Cells(k, 2) = n
    Cells(k, 3) = x
Next n
End Sub

```

Przykład 2. Metodą bisekcji i metodą stycznych (Newtona) wyznaczyć pierwiastki równania

$$\sin x + x^3 + 1 = 0.$$

Analiza wykresu funkcji $f(x) = \sin(x) + x^3 + 1$ sugeruje, że szukany pierwiastek znajduje się w przedziale $[-1, 0]$. Dlatego w metodzie bisekcji wybieramy $a = -1$, $b = 0$. Dla metody Newtona wybieramy jako punkt startu $x_0 = 0$ (choć jak widać z wykresu wybór $x_0 = -1$ byłby lepszy).



Rys. 2. Wykres funkcji $f(x)=\sin(x)+x^3+1$ na przedziale $[-1.3, 1.3]$. Wykres sugeruje, że miejsce zerowe znajduje się w przedziale $[-1, 0]$.

Do metody Newtona (3) potrzebujemy wyrażenia na pochodną, zatem obliczamy

$$f'(x) = (\sin(x) + x^3 + 1)' = \cos(x) + 3x^2,$$

co daje następujący ciąg przybliżeń

$$x_{n+1} = x_n - \frac{\sin(x_n) + x_n^3 + 1}{\cos(x_n) + 3x_n^2} \quad \text{dla } n = 0, 1, 2, \dots \quad (4)$$

Rozpoczynając iteracje od $x_0 = 0,0$ otrzymujemy (należy posłużyć się kalkulatorem lub arkuszem kalkulacyjnym – na przykład jak na **Rys. 3**):

$$\begin{aligned} x_1 &= x_0 - \frac{\sin(x_0) + x_0^3 + 1}{\cos(x_0) + 3x_0^2} = 0 - \frac{\sin(0) + 0^3 + 1}{\cos(0) + 3 \cdot 0^2} = -1, \\ x_2 &= x_1 - \frac{\sin(x_1) + x_1^3 + 1}{\cos(x_1) + 3x_1^2} = -1 - \frac{\sin(-1) + (-1)^3 + 1}{\cos(1) + 3 \cdot (-1)^2} = -0,7623 \\ x_3 &= x_2 - \frac{\sin(x_2) + x_2^3 + 1}{\cos(x_2) + 3x_2^2} = -0,7623 - \frac{\sin(-0,7623) + (-0,7623)^3 + 1}{\cos(-0,7623) + 3 \cdot (-0,7623)^2} = -0,7081, \\ &\dots \end{aligned}$$

Wyniki zestawiamy w tabeli.

Metoda	Kolejne iteracje						
	1	2	3	4	5	6	7
Bisekcji	-0,5	-0,750000	-0,625000	-0,687500	-0,718750	-0,703125	-0,710938
Newtona	-1,0	-0,762317	-0,708153	-0,705698	-0,705694	-0,705694	-0,705694

	A	B	C	D	E	F
1	Równanie:		$f(x)=\sin(x)+x^3+1, f'(x)=\cos(x)+3x^2$ $x_{n+1}=x_n - f(x_n)/f'(x_n)$			
2	$\sin(x) + x^3 + 1 = 0$					
3	Iteracje metody		$=B6 - (\text{SIN}(B6)+B6^3+1)/(\text{COS}(B6)+3*B6^2)$			
4	Newtona					
5	n	x_n				
6	0	0,000000				
7	1	-1,000000				
8	2	-0,762317				
9	3	-0,708153				
10	4	-0,705698				
11	5	-0,705694				
12	6	-0,705694				
13	7	-0,705694				
14	8	-0,705694				
15	9	-0,705694				
16	10	-0,705694				

Rys. 3. Realizacja metody Newtona dla równania $\sin(x)+x^3+1=0$ w oparciu o formuły arkusza kalkulacyjnego.

Przykład 3. (metoda Herona)^[3] Z metody Newtona wynika bardzo szybki algorytm obliczania pierwiastków kwadratowych. Aby obliczyć pierwiastek kwadratowy z liczby $c \in \mathbb{R}_+$ musimy tak naprawdę rozwiązać następujące równanie:

$$x^2 - c = 0, \quad (5)$$

czyli $f(x) = 0$ dla $f(x) = x^2 - c$. Teraz wystarczy zastosować wzór (3) na iteracje Newtona dla tej właśnie funkcji, wykorzystując pochodną $f'(x) = 2x$:

$$x_{n+1} = x_n - \frac{f(x_n)}{f'(x_n)} = x_n - \frac{x_n^2 - c}{2x_n} = \frac{2x_n^2 - x_n^2 + c}{2x_n} = \frac{x_n^2 + c}{2x_n} = \frac{1}{2} \left(x_n + \frac{c}{x_n} \right),$$

czyli formuła Herona

$$x_{n+1} = \frac{1}{2} \left(x_n + \frac{c}{x_n} \right). \quad (6)$$

Punkt startu na ogół przyjmujemy tak (choć w tym przypadku można zacząć iteracje od dowolnej liczby dodatniej, $x_0 > 0$):

$$x_0 = \begin{cases} 1 & \text{gdzie } 0 < c \leq 1, \\ c & \text{gdzie } c > 1. \end{cases} \quad (7)$$

Przykład 4. Obliczyć przybliżenie pierwiastka $\sqrt{5}$ metodą Herona.

³ Heron z Aleksandrii (ok. 10 – ok. 70 n.e.) – grecki uczonec, wynalazca i matematyk. Znany m.in. z konstrukcji pierwowzoru turbiny parowej (tzw. *bania Herona*) i wzoru na pole trójkąta poprzez tylko długości boków (*wzór Herona*). Jako pierwszy opisał metodę iteracyjną do obliczania pierwiastków kwadratowych liczb rzeczywistych.

Rozwiązanie:


$$x_0 = 5, \quad x_1 = \frac{1}{2}(x_0 + 5/x_0) = 3, \quad x_2 = \frac{1}{2}(x_1 + 5/x_1) = 7/3, \quad x_3 = \frac{1}{2}(x_2 + 5/x_2) = 47/21$$

$$x_4 = \frac{1}{2}(x_3 + 5/x_3) = 47/21, \quad x_5 = \frac{1}{2}(x_4 + 5/x_4) = 2207/987 \approx 2,23607.$$


Jak widać z wyników dokładność do pięciu cyfr po przecinku otrzymujemy już w piątej iteracji! Algorytm jest na tyle szybki, że można go stosować ręcznie (wykonując rachunki na papierze), gdyż wystarczą często dwie lub trzy iteracje, aby uzyskać dobre przybliżenie.

Przykład 5. (pH słabego kwasu HA w funkcji stężenia kwasu). Zależność pH od stężenia c_0 kwasu jednoprotownego wprowadzonego do roztworu wymaga w ogólnym przypadku rozwiązania równania trzeciego stopnia. Co prawda w podręcznikach podawane jest przeważnie równanie, które jest stopnia drugiego (gdy wyrażone przez stopień dysocjacji α , wtedy nazywane *prawem rozcieńczeń Ostwalda*), to jednak ściślejsze ujęcie – uwzględniające także jony oksoniowe (H^+ , a dokładniej H_3O^+) pochodzące z *autodysocjacji samej wody* – prowadzi do równania trzeciego stopnia.

Zakładamy więc, że mamy roztwór wodny słabego kwasu HA o stężeniu analitycznym c_0 . Reakcja dysocjacji w wodzie prowadzi do równowagi



(dokładniej: $HA + H_2O \rightleftharpoons H_3O^+ + A^-$) ze stałą równowagi K_a . Woda również ulega autojonizacji



(dokładniej: $H_2O + H_2O \rightleftharpoons H_3O^+ + OH^-$) ze stałą równowagi (iloczyn jonowy wody), K_w . Mamy więc dwa równania (wynikające z prawa działania mas) dla stężeń w stanie równowagi

$$\frac{[H^+][A^-]}{[HA]} = K_a, \quad (8)$$

$$[H^+][OH^-] = K_w,$$

gdzie wszystkie stężenia wyrażane są w $\text{mol} \cdot \text{dm}^{-3}$. Bilans ilości kwasu daje kolejne równanie

$$c_0 = [HA] + [A^-], \quad (9)$$

a bilans ładunku

$$[H^+] = [OH^-] + [A^-]. \quad (10)$$

Wprowadzamy oznaczenie na stężenie jonów wodorowych $x = [H^+]$ i mamy następujący układ równań:

$$\frac{x[A^-]}{[HA]} = K_a, \quad x[OH^-] = K_w, \quad (11)$$

$$c_0 = [HA] + [A^-], \quad x = [OH^-] + [A^-].$$

Metody Matematyczne w Procesach Przemysłowych, Notatki do wykładu,

Wykład 3, 4 (równowagi chemiczne, rozwiązywanie równań nieliniowych)

Z drugiego równania $[\text{OH}^-] = K_w / x$ i wstawiamy do czwartego: $x = K_w / x + [\text{A}^-]$. Z pierwszego wyliczamy $[\text{HA}^-] = x[\text{A}^-] / K_a$ i wstawiamy do trzeciego: $c_0 = x[\text{A}^-] / K_a + [\text{A}^-]$, skąd $[\text{A}^-] = c_0 / (1 + x / K_a)$. Po wstawieniu tego wyrażenia do równania na x (w pierwszym wierszu) otrzymujemy równanie na stężenie kationów wodorowych:

$$x = \frac{K_w}{x} + \frac{c_0 K_a}{x + K_a}. \quad (12)$$

Równanie to można przekształcić do równania wielomianowego trzeciego stopnia:

$$x^3 + K_a x^2 - (K_w + c_0 K_a)x - K_a K_w = 0. \quad (13)$$

Rozwiązując to równanie dla różnych stężeń $c_0 = [\text{HA}]_0$ wprowadzonego słabego kwasu HA otrzymujemy zależność $x = x(c_0)$, a w konsekwencji zależność $\text{pH} = -\log[\text{H}^+] = -\log x(c_0)$, czyli $\text{pH} = f(c_0)$ (pH jako funkcję stężenia kwasu). Na wykresie lepiej jednak przedstawiać zależność od logarytmu stężenia jonów wodorowych, tj. $\text{pH} = f(\log c_0)$ – patrz Rys. 4.

Równania (12) i (13) są oczywiście równoważne matematycznie, więc do numerycznego rozwiązywania można wziąć dowolne z nich. Testowe obliczenia zrealizowano zarówno dla sformułowania (12) jak i (13). Rozwiązywanie równania (12), które ma postać równania na punkt stały: $x = F(x)$, gdzie $F(x) = \frac{K_w}{x} + \frac{c_0 K_a}{x + K_a}$, wykonano *metodą iteracji prostych*:

$$x_{n+1} = F(x_n), \quad n = 0, 1, 2, \dots$$

Przykładowe obliczenia przeprowadzono dla kwasu octowego CH_3COOH ($K_a = 1,7378 \cdot 10^{-5}$) rozpuszczanego w wodzie w temperaturze 25 °C. W tych warunkach $K_w = 1,008 \cdot 10^{-14}$.

Na przykład dla $c_0 = 1,0$ oraz punktu startowego $x_0 = 0,01$ uzyskano wynik $\text{pH} = 2,3806$ (czyli $[\text{H}^+] = 0,00416264$) metodą Newtona (już przy piątej iteracji!). Metoda iteracji prostych dla punktu stałego (12) nawet po 50 iteracjach daje $\text{pH} = 2,03395$ (czyli $[\text{H}^+] = 0,00924797$) co należy uznać za wynik niewystarczający. Aby sporządzić wykres funkcji $\text{pH} = f(\log c_0)$ wykonano obliczenia na podstawie wzoru (13) dla analitycznych stężeń $c_0 = 10^0, 10^{-1}, \dots, 10^{-10}$. Poniżej są wyniki obliczeń.

$c_0 / (\text{mol} \cdot \text{dm}^{-3})$	10^0	10^{-1}	10^{-2}	10^{-3}	10^{-4}	10^{-5}	10^{-6}
$x = [\text{H}^+]$	$4,1626 \cdot 10^{-3}$	$1,3104 \cdot 10^{-3}$	$4,0852 \cdot 10^{-4}$	$1,2350 \cdot 10^{-4}$	$3,3911 \cdot 10^{-5}$	$7,1027 \cdot 10^{-6}$	$9,5832 \cdot 10^{-7}$
pH	2,3806	2,8826	3,3888	3,9083	4,4697	5,1485	6,0185
$c_0 / (\text{mol} \cdot \text{dm}^{-3})$	10^{-7}	10^{-8}	10^{-9}	10^{-10}			
$x = [\text{H}^+]$	$1,6150 \cdot 10^{-7}$	$1,0549 \cdot 10^{-7}$	$1,0089 \cdot 10^{-7}$	$1,0045 \cdot 10^{-7}$			
pH	6,7918	6,9768	6,9961	6,9980			

Dla porównania wyników dokładnych (w oparciu o wzór (12) lub (13)) policzmy jeszcze pH z tradycyjnego prawa rozcieńczeń Ostwalda – w którym nie uwzględnia się stężenia jonów wodorowych (oksoniowych) pochodzących z autodysocjacji wody. Można formalnie otrzymać to

równanie przyjmując w równaniu (12) lub (13) wartość $K_w = 0$. Mamy wtedy $x^3 + K_a x^2 - c_0 K_a x = 0$, czyli

$$x^2 + K_a x - c_0 K_a = 0. \quad (14)$$


Równanie to ma jedno dodatnie rozwiązanie $x = \frac{-K_a + \sqrt{K_a^2 + 4K_a c_0}}{2}$, zatem

$$\text{pH} = -\log \frac{-K_a + \sqrt{K_a^2 + 4K_a c_0}}{2}. \quad (15)$$

Dla porównania przedstawiamy wyniki pH obliczone wg modelu dokładnego – wzór (13) – i przybliżonego (prawo Ostwalda) – wzór (15) – w tabelce i na wykresie.

Tabela 1. Wyniki obliczeń pH wodnego roztworu kwasu octowego w zależności od stężenia kwasu (pierwszy wiersz). W drugim wierszu są wartości pH obliczone z uwzględnieniem jonów H^+ pochodzących z autodysocjacji wody. Trzeci wiersz zawiera pH obliczone z pominięciem jonów H^+ pochodzących od wody. Dla małych stężeń ($<10^{-6}$) wyniki zaczynają się wyraźnie różnić. Co więcej dla stężeń $c_0 \leq 10^{-7}$ pH obliczone uproszczonym wzorem (15) wskazuje na zasadowy odczyn (!) – co jest oczywiście wynikiem błędnym. Wartości te są zaznaczone kolorem czerwonym.

$c_0 / (\text{mol} \cdot \text{dm}^{-3})$	10^0	10^{-1}	10^{-2}	10^{-3}	10^{-4}	10^{-5}	10^{-6}	10^{-7}	10^{-8}	10^{-9}	10^{-10}
pH (wg (13))	2,3806	2,8826	3,3888	3,9083	4,4697	5,1485	6,0185	6,7918	6,9768	6,9961	6,9980
pH (wg (14))	2,3806	2,8826	3,3888	3,9083	4,4697	5,1486	6,0230	7,0025	8,0002	9,0000	10,0000



Rys. 4. Zależność pH roztworu słabego kwasu od stężenia. Rozważanym kwasem jest kwas octowy ($K_a = 1,74 \cdot 10^{-5}$). Więcej szczegółów – patrz opis tabeli powyżej.

Przykład 6. Obliczymy skład równowagowy dla reakcji syntezy amoniaku. Problem ten prowadzi do równania algebraicznego czwartego stopnia, które rozwiążemy numerycznie.

Reakcję syntezy amoniaku z pierwiastków opisuje równanie



Z termodynamiki chemicznej wiemy, że w stanie równowagi stężenia (a dokładniej – aktywności chemiczne) reagentów spełniają równanie

$$\frac{a_{\text{NH}_3}^2}{a_{\text{N}_2} a_{\text{H}_2}^3} = K, \quad (17)$$

gdzie stała równowagi K (jest bezwymiarowa) zależy tylko od temperatury (dlatego czasami piszemy $K = K(T)$), a a_X oznacza aktywność reagenta X . Stała równowagi tej reakcji może być obliczona na podstawie entalpii tworzenia reagentów i wynosi $K_p = 4,34 \cdot 10^{-3}$ w temperaturze 300° , jeżeli przyjmiemy, że aktywności reagentów gazowych są równe ich ciśnieniom cząstkowym (parcjalnemu). Zakładając, że gazy zachowują się jak gaz doskonały, to aktywności możemy zastąpić ciśnieniami cząstkowymi ($a_X = p_X / p^\ominus$) i przybliżona postać równania równowagi jest następująca

$$\frac{p_{\text{NH}_3}^2 (p^\ominus)^2}{p_{\text{N}_2} p_{\text{H}_2}^3} = K_p = 4,34 \cdot 10^{-3},$$

gdzie $p^\ominus = 10^5 \text{ Pa}$ (ciśnienie standardowe). Wyrażając ciśnienie przez liczbę moli w oparciu równania stanu gazu doskonałego (równanie Clapeyrona) $pV = nRT \Rightarrow p = nRT / V$ otrzymujemy

$$\begin{aligned} K_p &= \frac{p_{\text{NH}_3}^2 (p^\ominus)^2}{p_{\text{N}_2} p_{\text{H}_2}^3} = \frac{(n_{\text{NH}_3} RT / V)^2 (p^\ominus)^2}{(n_{\text{N}_2} RT / V)(n_{\text{H}_2} RT / V)^3} = \frac{n_{\text{NH}_3}^2}{n_{\text{N}_2} n_{\text{H}_2}^3} \frac{(RT)^2}{RT (RT)^3} \frac{V^{-2}}{V^{-1} V^{-3}} (p^\ominus)^2 \\ &= \frac{n_{\text{NH}_3}^2}{n_{\text{N}_2} n_{\text{H}_2}^3} \left(\frac{p^\ominus}{RT} \right)^2 V^2 = \frac{n_{\text{NH}_3}^2}{n_{\text{N}_2} n_{\text{H}_2}^3} \left(\frac{p^\ominus V}{RT} \right)^2, \end{aligned} \quad (18)$$

skąd po podstawieniu wartości $p^\ominus, R, T$ otrzymamy

$$\frac{n_{\text{NH}_3}^2}{n_{\text{N}_2} n_{\text{H}_2}^3} = \left(\frac{RT}{V p^\ominus} \right)^2 K_p = \frac{(8,3144 \cdot 298)^2}{(10^5)^2} 4,34 \cdot 10^{-3} V^{-2} = \frac{2,7 \cdot 10^{-6}}{V^2} \quad (19)$$

gdzie V to objętość reaktora, w którym zachodzi synteza.

Dla przykładu weźmy dane początkowe: $n_{\text{N}_2}^0 = 10 \text{ mol}$, $n_{\text{H}_2}^0 = 30 \text{ mol}$. Ile moli amoniaku uzyskamy, gdy reakcja osiągnie stan równowagi, a objętość reaktora wynosi $V = 1 \text{ m}^3$?

Oznaczmy dla wygody przez x liczbę moli amoniaku (produkt) w stanie równowagi: $x = n_{\text{NH}_3}$. Z bilansu masy (równanie reakcji chemicznej (16)) w stanie równowagi mamy:

$$n_{\text{N}_2} = n_{\text{N}_2}^0 - \frac{1}{2} x, \quad n_{\text{H}_2} = n_{\text{H}_2}^0 - \frac{3}{2} x,$$

czyli $n_{\text{N}_2} = 10 - \frac{1}{2} x$, $n_{\text{H}_2} = 30 - \frac{3}{2} x$. Podstawiając te zależności oraz $V = 1 \text{ m}^3$ do równania (18) otrzymamy

$$\frac{x^2}{(10 - \frac{1}{2} x)(30 - \frac{3}{2} x)^3} = 2,7 \cdot 10^{-6} / V^2, \quad (20)$$

czyli

$$x^2 - (2,7 \cdot 10^{-6} / V^2)(10 - \frac{1}{2} x)(30 - \frac{3}{2} x)^3 = 0. \quad (21)$$


Metody Matematyczne w Procesach Przemysłowych, Notatki do wykładu,

Wykład 3, 4 (równowagi chemiczne, rozwiązywanie równań nieliniowych)

W szczególności dla $V = 1 \text{ m}^3$ powstaje równanie $x^2 = 2,7 \times 10^{-6} (10 - \frac{1}{2}x)(30 - \frac{3}{2}x)^3$. Jest to równanie algebraiczne stopnia 4. Stosując metodę Newtona znaleziono cztery rozwiązania rzeczywiste:

$$x_1 = 0,7878, \quad x_2 = -427,55, \quad x_3 = -0,9355, \quad x_4 = 5007,6980.$$


Drugie, trzecie i czwarte rozwiązania są oczywiście нефизyczne: ujemne liczby moli amoniaku, a czwarte daje ujemną liczbę moli wodoru: $n_{\text{H}_2} = 10 - \frac{1}{2}x = 10 - \frac{1}{2} \cdot 5007,6980 = -2493,85$. Jedynym poprawnym fizycznie rozwiązaniem jest $n_{\text{NH}_3} = x_1 = 0,7878$, skąd $n_{\text{N}_2} = 9,60$, $n_{\text{H}_2} = 28,82$. Ze stechiometrii równania syntezy amoniaku wynika, że maksymalna liczba moli NH_3 , która wynika z samego bilansu masy jest równa $2 \cdot n_{\text{N}_2}^0$, czyli w analizowanym przypadku $n_{\text{NH}_3(\text{max})} = 20$. Zatem wydajność reakcji $\eta = n_{\text{NH}_3} / n_{\text{NH}_3(\text{max})} = 3,9\%$. Gdy początkowe stężenia substratów będą na przykład 100-krotnie większe (a zatem ciśnienie będzie większe), to $n_{\text{NH}_3} = 1238,33$ oraz $\eta = 62\%$. Równowaga przesuną się w prawo – patrząc na równanie chemiczne syntezy NH_3 widzimy, że jest to zgodne z *regułą przekory* (regułą Le Chateliera).



Rys. 1. Na wykresie pokazana jest zależność rzeczywistej wydajności reakcji (η (%)) syntezy amoniaku z pierwiastków w zależności od ciśnienia przy stałej temperaturze $T=25^\circ\text{C}$. Ciśnienie w jednostkach umownych. Dane w tabelce (pokazane częściowo) są zilustrowane na wykresie obok, zostały obliczone w programie Ms Excel.

Metoda 1 (Użycie dodatku Solver). Aby otrzymać zależność jak na **Rys. 1** należy w równaniu (20) podstawiać różne wartości objętości V , nie zmieniając pozostałych współczynników, które zależą od stałych fizycznych, temperatury i warunków początkowych (liczba moli $n_{\text{H}_2}^0$, $n_{\text{N}_2}^0$). Tak więc zmiana V oznacza zmianę ciśnienia, gdyż $p \sim 1/V$. Dla prostoty przyjęto na wykresie w umownych jednostkach $p = 1/V$. Kluczowymi wartościami w tabeli są liczby w kolumnie $[\text{NH}_3]_{\text{rów}}$, które są otrzymane poprzez rozwiązywanie równania (20). Najprostszym sposobem uzyskania rozwiązania jest wykorzystanie dodatku Solver, ale jest to uciążliwe i czasochłonne, gdyż musimy wielokrotnie ręcznie zmieniać V , uruchamiać od nowa Solver, a następnie wynik kopiować do tabelki. W przypadku wykresu na **Rys. 1** należało tę procedurę wykonać 28 razy.

Metody Matematyczne w Procesach Przemysłowych, Notatki do wykładu,
Wykład 3, 4 (równowagi chemiczne, rozwiązywanie równań nieliniowych)



Rys. 2. Realizacja obliczeń stężenia amoniaku (Przykład 4) poprzez rozwiązywanie równania (21) przy użyciu dodatku Solver w arkuszu kalkulacyjnym Ms Excel.

Metoda 2 (program w języku VBA). Na rysunku Rys. 3 pokazana jest przykładowa realizacja rozwiązywania równania na równowagowe stężenie amoniaku w syntezie (16) przy pomocy procedury w języku VBA. Numeryczne rozwiązanie uzyskiwane jest metodą Newtona (3), gdzie

$$a = 0,0000027,$$

$$f(x) = x^2 - (a / V^2)(10 - 0,5x)(30 - 1,5x)^3, \quad (22)$$

$$Df = f'(x) = 2x + (a / V^2)(0,5(30 - 1,5x)^3 + 4,5(10 - 0,5x)(30 - 1,5x)^2).$$

Liczba iteracji metody Newtona wynosi 100. Zaletą tego podejścia w porównaniu do Metody 1 jest to, że program automatycznie realizuje obliczenia dla dowolnego zakresu zmiennej V. Jak widać w pierwszym obiegu pętli **while** mamy V = 8, a w każdym kolejnym objętość jest dzielona przez 1,5.

```
Const a = 0.0000027
Function f(x As Double, V As Double) As Double
    f = x ^ 2 - (a / V ^ 2) * (10 - 0.5 * x) * (30 - 1.5 * x) ^ 3
End Function
Function Df(x As Double, V As Double) As Double
    Df = 2 * x + (a / V ^ 2) * (0.5 * (30 - 1.5 * x) ^ 3 + 4.5 * (10 - 0.5 * x) * (30 - 1.5 * x) ^ 2)
End Function
Sub oblicz_NH3()
    Dim V As Double
    Dim x As Double
    Dim i As Integer
    Dim k As Integer

    V = 8
    x = 0.1
    i = 5
    While V > 0.0001
        For k = 1 To 100
            x = x - f(x, V) / Df(x, V)
        Next k

        Cells(i, 2) = V
        Cells(i, 3) = 1 / V
        Cells(i, 4) = x
        i = i + 1
        V = V / 1.5
    Wend
End Sub
```

Rys. 3. Realizacja obliczeń stężenia amoniaku (Przykład 4) poprzez rozwiązywanie równania (21) przy użyciu procedury w języku VBA w arkuszu kalkulacyjnym Ms Excel.

Przykład 7 (Równanie gazu van der Waalsa). Równanie van der Waalsa ma postać

$$p = \frac{nRT}{V - nb} - a \left(\frac{n}{V} \right)^2, \quad (23)$$

